

Program Example of RS-232C Communication

In this program, when inputting transmission data in command format from a computer keyboard, data returned from the Processor that conforms to the response format is displayed on the computer screen. Input the data to be transmitted from the start character "@" to the last piece of text data. After data input, FCS is calculated and transmitted with a terminator. If this program is not executed correctly, there will be an error in the transmission section; check the connection of communication cables, etc.

(This program is written in N88 BASIC. An NEC PC-9801 personal computer is used as the host system.)

```
1000 ' -----
1010 ' PROGRAM:K3N* Communication Sample Program K3T* Compatible)
1020 ' VERSION:1.00
1030 ' (C)Copyright OMRON Corporation 1997
1040 ' All Rights Reserved
1050 ' -----
1060 '
1070 ' ==== COM port settings (even parity, 7 data bits, 2 stop bits) ====
1080 OPEN"COM:E73:AS#1
1090 '
1100 *REPEAT
1110 ' ==== Transmission process ====
1120 ' --- Input transmission data. ---
1130 INPUT"SEND DATA:",SEND$
1140 '
1150 ' --- Jump to exit process if there is no input. ---
1160 IF SEND$="" THEN *EXIT
1170 '
1180 ' --- BCC calculation ---
1190 FCS=0
1200 FOR I=1 TO LEN(SEND$)
1210 FCS=FCS XOR ASC(MID$(SEND$,I,1))
1220 NEXT I
1230 FCS$=CHR$("0"+HEX$(FCS),2)
1240 '
1250 ' --- Transmission ---
1260 SDATA$=SEND$+FCS$+"*"+CHR$(13)
1270 PRINT#1,SDATA$;
1280 '
1290 ' ==== Reception process ====
1300 RDATA$="":TIMEOUT=0
1310 *LOOP
1320 ' --- No-response detection ---
```

```
1330 TIMEOUT=TIMEOUT+1
1340 IF TIMEOUT>20000 THEN RESP$="No Response":GOTO *REND
1350 IF LOC(1)=0 THEN *LOOP
1360 '
1370 ' --- End character discrimination (Read if it isn't end character.)
---
1380 RDATA$=RDATA$+INPUT$(LOC(1),#1)
1390 IF RIGHT$(RDATA$,1)<>CHR$(13) THEN *LOOP
1400 RESP$=RDATA$
1410 *REND
1420 '
1430 ' --- Display received data. ---
1440 PRINT"Response:";RESP$
1450 GOTO *REPEAT
1460 '
1470 *EXIT
1480 ' ==== Exit process ====
1490 CLOSE#1
1500 END
```